
clear 
 

// Definicao da funcao que implementa a equacao nao linear 
function [hdot]=tanque(t, h, Qe) 
hdot(1)=(-sqrt(rho*g*(h(1)-h(2))/R)+Qe(t))/S 
hdot(2)=(-sqrt(rho*g*h(2)/R)+sqrt(rho*g*(h(1)-h(2))/R))/S 
endfunction 
 
// Definicao da funcao que implementa a entrada Qe: 
function [u]=entrada(t) 
u=Qei; 
// supondo o exemplo, u=K1*sin(w*t)+K2*t^(-2) 
endfunction 
 
// Definir parametros: 
S=10; // [m^2] Area da secao transversal do reservatorio 
rho=1000; // [kg/m^3] massa especifica da agua 
g=10; // [m/s^2] aceleração da gravidade na superficie da Terra 
R=2*10^8; // [Pa/(m^3/s)^2] parametro que relaciona pressao e vazao 
ho=2; // [m] nivel do reservatorio em regime 
hi=0.1; // [m] nível adicional desejado 
Qei=sqrt(rho*g*(ho+hi)/R); // [m^3/s] vazao na entrada 
// Definir a condicao inicial: 

h0=[0;0]; // [m] nivel do reservatorio na condicao inicial 
// Definir o vetor t de instantes de tempo: 
t=0:10:40000; // vetor de tempo. Observe que t(1) eh o instante inicial 
// Comando que realiza a simulacao numérica: 
h=ode(h0,t(1),t,list(tanque,entrada)); // h eh o nivel do reservatorio [m] 
// Plotando o resultado em verde: 
plot2d(t,h(1,:),2) 
plot2d(t,h(2,:),1) 
legends(["h1","h2"],[2 1]) 
// Definindo uma variavel do tipo 'lista': 
T=list("Resposta transitoria do reservatorio","Tempo t [s]","Nivel h [m]"); 
// Colocando um titulo na figura e nomeando os eixos: 
xtitle(T(1),T(2),T(3)); 

 


